AD-753 842

DEPARTMENT OF DEFENSE LAND FALLOUT PREDICTION SYSTEM. VOLUME II: INITIAL CONDITIONS, SUPPLEMENT

Hillyer G. Norment

Mount Auburn Research Associates, Incorporated

Prepared for:

Defense Nuclear Agency

October 1972

DISTRIBUTED BY:

NTIS

National Technical Information Service
U. S. DEPARTMENT OF COMMERCE
5285 Port Royal Road, Springfield Va. 22151

DNA 1800-II
Supplement
October 1972

DEPARTMENT OF DEFENSE LAND FALLOUT PREDICTION SYSTEM

VOLUME II

INITIAL CONDITIONS
Supplement

Prepared by

H. G. Norment

MT. AUBURN RESEARCH ASSOCIATES, INC.
385 Elliot Street
Newton, Massachusetts 02164

This work was supported by the Defense Nuclear Agency
under NWED/Subtask PD068-04

Sponsored by

Defense Nuclear Agency
Washington, D. C. 20305

Contract No. DASA01-71-C-0118

Approved for public release;
distribution unlimited.

33

Unclassified
Security Classification

DOCUMENT CONTROL DATA - R & D

(Security classification of title, body of abstract and indexing annotation must be entered when the overall report is classified)

1. ORIGINATING ACTIVITY (Corporate author) Mt. Auburn Research Associates, Inc. 385 Elliot Street Newton, Massachusetts 02164	2a. REPORT SECURITY CLASSIFICATION Unclassified 2b. GROUP

3. REPORT TITLE
"Department of Defense Land Fallout Prediction System. Volume II: INITIAL CONDITIONS, Supplement"

4. DESCRIPTIVE NOTES (Type of report and inclusive dates)
Final Report

5. AUTHOR(S) (First name, middle initial, last name)
Hillyer G. Norment

6. REPORT DATE	7a. TOTAL NO. OF PAGES	7b. NO. OF REFS
October 1972	37	5

8a. CONTRACT OR GRANT NO. DASA01-71-C-0118 b. PROJECT NO. NWED:XAXP c. Task & Subtask: D068 d. Work Unit: 04	9a. ORIGINATOR'S REPORT NUMBER(S) 9b. OTHER REPORT NO(S) (Any other numbers that may be assigned this report) DASA 1800-II Supplement

10. DISTRIBUTION STATEMENT

Approved for public release; distribution unlimited.

11. SUPPLEMENTARY NOTES	12. SPONSORING MILITARY ACTIVITY Director Defense Nuclear Agency Washington, D. C. 20305

13. ABSTRACT

The DELFIC Initial Conditions Module code has been revised to meet the requirements of the new DELFIC fallout prediction system. This documentation supplement describes the revised code. Discussion of the revised code emphasizes particle size distributions. The code can accept parameters that define lognormal or power-law distributions, or it can accept a distribution in tabular form. Details necessary for use of the code are presented. FORTRAN statement listings of revised subroutines are included.

Ia

DD FORM 1 NOV 65 1473 REPLACES DD FORM 1473, 1 JAN 64, WHICH IS OBSOLETE FOR ARMY USE.

Unclassified
Security Classification

Unclassified

Security Classification

14. KEY WORDS	LINK A		LINK B		LINK C	
	ROLE	WT	ROLE	WT	ROLE	WT
Nuclear weapon effects						
Radioactivity						
Fallout prediction						
DELFIC						

Ib

Unclassified

DNA 1800-II
Supplement

October 1972

DEPARTMENT OF DEFENSE LAND FALLOUT PREDICTION SYSTEM

VOLUME II

INITIAL CONDITIONS
Supplement

Prepared by

H. G. Norment

MT. AUBURN RESEARCH ASSOCIATES, INC.
385 Elliot Street
Newton, Massachusetts 02164

This work was supported by the Defense Nuclear Agency
under NWED/Subtask PD068-04

Sponsored by

Defense Nuclear Agency
Washington, D. C. 20305

Contract No. DASA01-71-C-0118

Approved for public release;
distribution unlimited.

II

TABLE OF CONTENTS

1. INTRODUCTION

Since publication of DASA-1800-II[1] in 1966, the Defense Land Fallout Interpretative Code (DELFIC) has undergone substantial revision in all of its modules. These revisions have created some new demands on the Initial Conditions Module (ICM) and removed some old restrictions. Of most direct consequence to the ICM are changes in the Cloud Rise Module (CRM)[2].

The new CRM accounts for wind shear effects on the cloud rise dynamics. Therefore, shot-time winds above ground zero are input via the ICM rather than via the Cloud Rise-Transport Interface Module (CRTIM) as was done originally. The old CRM could accept no more than forty particle size classes, and the size class structure was rigidly prescribed. These restrictions have been relaxed in the new CRM, and the ICM has been revised accordingly. In addition, the ICM has been given a capability to accept parameters that define a power-law particle size distribution function. From these parameters, it constructs a particle size class table with a user-specified number of entries.

Subroutines LINK1 and DSTBN have been revised, and a new subroutine, SHWIND, which is called by LINK1, has been created. Subroutines MASS, TEMP, TIME, and VAPOR remain unchanged.

Subroutine LINK1 is the ICM executive program. Subroutine DSTBN constructs particle size class tables for lognormal and power-law particle distributions. Subroutine SHWIND reads in the shot-time winds above ground zero.

The logic of the ICM consists of a card input, which is described in Table 1, followed by serial exercise of the subordinate subroutines. Adequate detailed documentation is provided by the FORTRAN statement listings.

Use of the ICM is quite simple with one exception: definition of particle size distributions. Therefore, the bulk of this supplement is devoted to discussions of particle size distributions.

2. THE LOGNORMAL DISTRIBUTION

2.1 Fundamentals

A variable x is said to be normally distributed if the probability of its occurrence in the range x to x + dx is given by

$$dN(x|\mu,\sigma^2) = \frac{1}{\sigma\sqrt{2\pi}} \exp\left[-\frac{1}{2}\left(\frac{x-\mu}{\sigma}\right)^2\right] dx \ , \qquad (1)$$

where μ is the mean value of x and σ^2 is the variance of x. The square root of the variance, σ, is called the standard deviation.

To define a lognormal distribution, we make the transformation

$$x = \ln(y) \ . \qquad (2)$$

In terms of the variable y Eq. (1) becomes

$$d\Lambda(y|\mu,\sigma^2) = \frac{1}{\sigma\sqrt{2\pi}} \exp\left[-\frac{1}{2}\left(\frac{\ln y - \mu}{\sigma}\right)^2\right] d(\ln y) \ , \qquad (3)$$

and y is said to be lognormally distributed.(3)

Some statistical properties of y are as follows:

$$\text{mean}(y) = \exp(\mu + \tfrac{1}{2}\sigma^2) \qquad (4)$$

$$\text{median}(y) = \exp(\mu) \qquad (5)$$

$$\text{mode}(y) = \exp(\mu - \sigma^2) \qquad (6)$$

$$\text{variance}(y) = \left[\exp(\sigma^2) - 1\right] \exp(2\mu + \sigma^2) \ . \qquad (7)$$

Preceding page blank

Let $\underline{y}$ and s be the geometric mean and geometric standard deviation of y. Then

$$\underline{y} = \text{median}(y) = \exp(\mu) \tag{8}$$

and

$$s = \exp(\sigma) \ . \tag{9}$$

Let λ_j' be the j-th moment of $\Lambda(y|\mu,\sigma^2)$ about the origin. Then by definition

$$\lambda_j' = \int_0^\infty y^j \, d\Lambda(y|\mu,\sigma^2) \quad , \tag{10}$$

and from the properties of the normal distribution it follows that

$$\lambda_j' = \exp(j\mu + \frac{1}{2} j^2\sigma^2) \quad . \tag{11}$$

A feature that distinguishes the lognormal distribution from the normal distribution is the existance of moment distributions. The j-th moment distribution is defined as

$$\Lambda(y|\mu,\sigma^2)_j = \frac{1}{\lambda_j'} \int_0^y t^j d\Lambda(t|\mu,\sigma^2) \quad , \tag{12}$$

which can be shown to be[3]

$$\Lambda(y|\mu,\sigma^2)_j = \Lambda(y|\mu + j\sigma^2,\sigma^2) \quad . \tag{13}$$

The moment distributions provide simple relationships between lognormal distributions of number, surface area, and volume of particles with respect to their diameters.

2.2 Application to Particle Distributions

In discussions of lognormal particle distributions, confusion frequently arises because distinction is not clearly made between μ and $\underline{y}$ and between σ and s. Since particular values of μ and σ depend on the base of the logarithms used, we have chosen to confine our discussions in the DELFIC documentation to parameters in the form of $\underline{y}$ and s.

Suppose that we have plotted cumulative numbers of particles versus diameter on log-probability graph paper and have obtained the curve shown in Figure 1. This straight-line curve indicates that the distribution of particle number with respect to diameter, D, is lognormal. Thus, D is equivalent to y in Eq. (3), and from Eqs. (8) and (9) we have

$$\underline{y} = D_{50}$$

and

$$s = D_{84.13}/D_{50}$$

These are the quantities DMEAN and SD, respectively, that are required as input to subroutine LINK1, and that are printed by LINK1. DMEAN is expressed in units of micrometers. (In words, DMEAN is the median particle diameter in the distribution of numbers of particles with respect to their diameters.) If the user specifies a lognormal distribution but does not input values for DMEAN and SD, the program supplies the values[1]:



FIGURE 1. CUMULATIVE FREQUENCY GRAPH OF LOGNORMALLY DISTRIBUTED PARTICLES

$$DMEAN = \underline{y} = 0.407\ \mu m$$

$$SD = s = 4.0$$

As noted above, the properties of the moment distributions are useful in interrelating distributions of particle number, surface area, and volume with respect to particle diameter. This is because the number distribution is the zeroth moment distribution with respect to diameter, surface area is distributed via the second moment distribution, and volume is distributed via the third moment distribution. Thus, if we assume spherical particles and if the parameters μ and σ are known for either the particle number, or particle area, or particle volume distribution with diameter, then the other distributions can be determined from the equations below. The parameter σ is the same for all three distributions. If we use N, S, and V as subscripts to denote number, surface area, and volume, respectively, we have from Eq. (13)

$$\mu_S = \mu_N + 2\sigma^2$$

$$\mu_V = \mu_N + 3\sigma^2$$

where μ and σ are related to $\underline{y}$ and s by Eqs. (8) and (9).

If base 10 logarithms are used instead of natural logarithms, we distinguish the distribution parameters by use of primes, μ_N' and σ', and the relations become

$$\mu_S' = \mu_N' + 2\ell n(10)(\sigma')^2$$

$$\mu_V' + \mu_N' + 3\ell n(10)(\sigma')^2$$

where $\ell n(10) = 2.3026$.

The distribution of particle mass with respect to diameter is taken to be equivalent to the volume distribution with respect to diameter. This implies that all particles have the same density.

2.3 Particle Size Class Tables

For computation purposes, the continuous lognormal distribution is replaced by a histogram. The computer program, via subroutine DSTBN, does this automatically by use of the distribution parameters and the number of size classes, NDSTR, which is input by the user.

The user specifies parameters DMEAN, SD, and NDSTR. From these, the parameters μ_N, σ, and μ_V are determined via

$$\mu_N = \ell n(\text{DMEAN})$$

$$\sigma = \ell n(\text{SD})$$

$$\mu_V = \mu_N + 3\sigma^2 \quad .$$

Define the normal distribution function argument x as

$$x = \frac{\ell n(D) - \mu_V}{\sigma}$$

where D is particle diameter. Then

$$N(x) = \frac{1}{\sqrt{2\pi}} \int_{-\infty}^{x} \exp(-t^2/2)dt$$

Subroutine DSTBN constructs the pa. cle size class table (i.e., histogram) as follows. Each size class contains a constant volume fraction, ΔN_V, of

$$\Delta N_V = 1/NDSTR \quad .$$

Let D_i, $i = 1,2.....,NDSTR$, be the upper (i.e. the larger particle) boundary diameter of the i-th particle size class. The table is ordered with the largest particles in the first size class, and so on. Then, for the i-th size class

$$N(x_i) = i\Delta N_V$$

and

$$\ell n(D_{i+1}) = x_i\sigma + \mu_V \quad .$$

The upper boundary of the first size class, D_1, and the lower boundary of the last size class, $D_{NDSTR+1}$, are special cases. These are taken to be the diameters at $\Delta N_V/2$ and $1-\Delta N_V/2$, respectively. That is,

$$N(x_1) = \frac{\Delta N_V}{2}$$

and

$$N(x_{NDSTR+1}) = 1 - \frac{\Delta N_V}{2} \; .$$

In these calculations x is determined from given N(x) via equation 26.2.23 of Reference 4.

The central particle diameter for the i-th class, d_i, is given by

$$d_i = \sqrt{D_i D_{i+1}} \quad .$$

If NDSTR = 1, a single size class is created with

$$D_1 = (DMEAN) * (5.0 * SD)$$

$$D_2 = (DMEAN) / (5.0 * SD)$$

and

$$d_1 = DMEAN \quad .$$

3. THE POWER-LAW DISTRIBUTION

3.1 Fundamentals

Mathematically speaking, power-law distributions are meaningless since distribution functions cannot be defined for them. This is because the power-law function is not properly bounded for zero argument Freiling has shown that fallout particle distributions that have been represented by power-law functions can equally well be fitted by lognormal distribution functions.[5] The implication of Freiling's work is that power-law distributions would be more accurately described as truncated lognormal distributions. Nevertheless, power-law distributions frequently are useful in fallout work.

Define the power-law frequency as

$$df(D|k,X) = kD^{-X}dD \ , \qquad (14)$$

where $df(D|k,X)$ is the number of particles in the diameter range D to D + dD. If we assume spherical particles with constant density, ρ, we have

$$dF\left(D\left|\frac{\pi\rho k}{6M},X\right.\right) = \frac{\pi\rho k D^{3-X}}{6M} \quad dD \ , \qquad (15)$$

where $dF\left(D\left|\frac{\pi\rho k}{6M},X\right.\right)$ is the fraction of the total fallout mass, M, in the diameter range D to D + dD.

The mass fraction of particles in the macro-range from D_i to D_j is obtained by integration of Eq. (15) between these limits to give

$$\Delta F = \frac{\pi\rho k}{6M(4-X)}\left(D_j^{4-X} - D_i^{4-X}\right), \qquad 0 < X < 4 \ . \qquad (16)$$

3.2 Particle Data Analysis

Suppose that we have obtained a sample of fallout particles. We weigh the sample to obtain M (kg), and we size the sample into N fractions, the i-th fraction containing particles in the diameter range ΔD_i centered on D_i (meters). We weigh each fraction and obtain the mass fractions ΔF_i. We determine that the average particle density is ρ (kg/m^3).

To obtain the power law distribution parameters k and X, we plot $\log(\Delta F_i/\Delta D_i)$ versus $\log(D_i)$. A straight line is fitted to the data. From Eq. (15), we see that the intercept and slope are

$$\text{intercept} = c = \log\left(\frac{\pi\rho k}{6M}\right) ,$$

and

$$\text{slope} = m = 3 - X .$$

Then

$$X = 3 - m$$

and

$$k = \frac{6M}{\pi\rho} \log^{-1}(c) .$$

When X and k are determined from M expressed in kilograms, D and ΔD in meters, and ρ in kg/m^3, they can be input to subroutine LINK1 as EXPO and CAY, respectively.

3.3 Particle Size Class Tables

For use in fallout calculations, subroutine DSTBN creates a histogram representation of the power law distribution. The histogram is comprised of NDSTR particle size classes, where NDSTR is specified by the user. The mass fraction in each size class, ΔF, is the constant

$$\Delta F = 1/NDSTR \quad .$$

Let D_i be the upper (i.e. larger particle) boundary of the i-th particle size class. The table is ordered with the largest particles in the first class, and so on. Then the smallest particles are contained in the NDSTRth class. If we assume that the smallest particle in this class is much smaller than D_{NDSTR}, we see from Eq. (16) that

$$D_{NDSTR}^{4-X} = \frac{6M(4-X)}{\pi\rho k} \Delta F \quad .$$

By recursive use of this relation with Eq. (16), we find that

$$D_i = (NDSTR - i+1)^{\frac{1}{4-X}} D_{NDSTR} \quad .$$

Size class central diameters, d_i, are

$$d_i = \sqrt{D_i D_{i+1}} \quad .$$

To establish a central and lower boundary diameter for the NDSTRth class, we say that

$$d_{NDSTR} = \left(\frac{1}{2}\right)^{\frac{1}{4-X}} D_{NDSTR}$$

and

$$D_{NDSTR+1} = (d_{NDSTR})^2/D_{NDSTR} \quad .$$

4. TABULAR DISTRIBUTIONS

4.1 Particle Size Class Tables

If the user so desires, he can input his particle size distribution in histogram form with NDSTR size classes. The table of size classes must be arranged in descending order of particle diameter. Each size class is defined in the input by its upper (i.e. larger particle) boundary diameter, D_i, and mass fraction, ΔF_i. These two data are punched on a separate card for each size class. The last card in the deck contains the lower boundary diameter of the NDSTRth size class. Central particle diameters, d_i, are computed to be

$$d_i = (D_i + D_{i+1})/2 .$$

Preceding page blank

5. USER INFORMATION

5.1 Card Input

The ICM card input is described in Table I. This table and the discussions in Sections 2.2, 2.3, 3.2, and 4.1 provide adequate information for use of the code.

5.2 Output

Though the printed output has been modified somewhat, the example output presented in DASA-1800-II is still satisfactory.

Communication with the Cloud Rise Module is via COMMON/SET1/. The contents of COMMON/SET1/ is described in Table 2.2 of DASA-1800-III (Revised).[2]

Preceding page blank

TABLE I

CARD INPUTS TO THE INITIAL CONDITIONS MODULE

Card Number	Contents	Variable Names and FORMATS
1	ICM Run Identifier	DETID(J), J=1,12 (12A6)
2	Control integer to specify particle size distribution type: 1 lognormal 2 power-law 3 tabular	IDISTR (I5)
3	Number of particle size classes	NDSTR (I5)
4(a)*	(For lognormal particle size distribution) Explosion yield (KT), height of burst above GZ(m), soil class indicator: 1.0 for siliceous 2.0 for calcareous, median particle diameter (μm), geometric standard deviation of the particle size distribution, and particle density (g/cm^3). (See Section 2.2.)	W, HEIGHT, USOIL, DMEAN, SD, DNS (6F10.3)
4(b)*	(For power-law particle size distribution) Yield (KT), height of burst (m), soil class indicator (see above), exponent in the particle size distribution frequency function, coefficient in the particle size distribution frequency function, particle density (g/cm^3). (See Section 3.2.)	W, HEIGHT, USOIL, EXPO, CAY, DNS (6F10.3)
4(c)*	(For a tabular particle size distribution) Yield (KT), height of burst (m), soil class indicator (see above), particle density (g/cm^3).	W, HEIGHT, USOIL, DNS (4F10.3)
$4(c)_1^{**}$	A table of upper boundary particle diameters (μm) and mass fractions	DIAM(J), FMASS(J), J=1,NDSTR (2E12.5)
. . .	. .	. .
$4(c)_{NDSTR+1}^{**}$	The lower boundary diameter (μm) of the last particle size class. (See Section 4.1.)	DIAM(NDSTR+1) (E12.5)

Table I (continued)

Card Number	Contents	Variable Names and FORMATS
5	Number of entires in the wind data table	NHODO (I5)
6***	For each entry in the wind data table: altitude (m, relative to msl), x component of wind (m/sec), y component of wind (m/sec)	ZV(J), VX(J), VY(J), J=1, NHODO (3F12.3)

* One of the cards 4(a), 4(b), or 4(c) is read according to whether IDISTR is 1, 2, or 3.

** These cards are read only for a tabular distribution.

*** These cards are read only if NHODO > 0.

6. FORTRAN STATEMENT LISTINGS

Complete FORTRAN statement listings are given for the following subroutines. These subroutines are operational on the UNIVAC 1108.

The machine used to prepare these listings prints a # symbol to represent a 4-8 punch; this symbol should be an apostrophe ('). In FORMAT and DATA statements, the apostrophe is used to define Hollerith character fields.

Preceding page blank

```
      SUBROUTINE LINK1
C     INITIAL CONDITIONS (FIREBALL) MODULE
C     MT. AUBURN RESEARCH ASSOCIATES        JANUARY 1972
C
C ******************************************************************
C
C     PROGRAM TO DETERMINE THE INITIAL CONDITIONS SPECIFICATIONS OF
C     TIME, TEMPERATURE, TOTAL SOIL MASS, FRACTION OF THE SOIL BURDEN IN
C     THE VAPOR PHASE, AND THE SIZE FREQUENCY DISTRIBUTION OF THE
C     CONDENSED PHASE SOIL
C
C     THE FIRST CARD CONTAINS ANY ARBITRARY ALPHANUMERIC IDENTIFICATION.
C
C     OTHER INPUT PARAMETERS ARE - TEST PARAMETER (IDISTR) TO DETERMINE
C     IF THE PARTICLE SIZE FREQUENCY DISTRIBUTION IS LOG-NORMAL, POWER
C     LAW, OR TABULAR, YIELD IN KILOTONS, HEIGHT(DEPTH) OF BURST IN
C     METERS, A SOIL TYPE INDICATOR, FALLOUT PARTICLE DENSITY(GM/CM**3),
C     MEAN(MICROMETERS) AND STANDARD DEVIATION FOR A LOG-NORMAL PARTICLE
C     SIZE FREQUENCY DISTRIBUTION, THE NUMBER OF PARTICLE SIZE CLASSES
C     IN THE PARTICLE SIZE FREQUENCY DISTRIBUTION. IF EITHER A TABULAR
C     OR POWER LAW DISTRIBUTION IS USED , THE MEAN AND STANDARD
C     DEVIATION ARE NOT CALLED FOR SINCE THEY DO NOT APPLY.    IF A
C     LOG-NORMAL DISTRIBUTION IS TO BE SUPPLIED BY THE PROGRAM, THE
C     MEAN AND STANDARD DEVIATION FIELDS ARE LEFT BLANK.
C     SHOT TIME WINDS ABOVE GZ ALSO ARE INPUT.  THESE ARE USED TO
C     COMPUTE WIND SHEAR EFFECTS ON CLOUD RISE AND FALLOUT ADVECTION
C     DURING THE CLOUD RISE TIME INTERVAL.
C
C     FOR UNDERGROUND BURSTS INPUT DEPTH OF BURST AS A NEGATIVE NUMBER
C
C     THE OUTPUT UNITS ARE MASS IN KILOGRAMS, LENGTH IN METERS, TIME IN
C     SECONDS, TEMPERATURE IN DEGREES KELVIN, YIELD IN KILOTONS,
C     DISTRIBUTION PARAMETERS IN MICRONS
C
C ***********************  GLOSSARY  *********************************
C
C     CAY          COEFFICIENT OF THE FREQUENCY FUNCTION FOR THE POWER
C                  LAW PARTICLE SIZE FREQUENCY DISTRIBUTION
C     DETID(I)     INITIAL CONDITIONS IDENTIFICATION ARRAY
C     DIAM(I)      ARRAY(201), UPPER BOUNDARY OF THE I-TH PARTICLE SIZE
C                  CLASS.  THE LAST ENTRY IN THE DIAM ARRAY IS THE LOWER
C                  BOUNDARY OF THE LAST(SMALLEST) PARTICLE SIZE CLASS.
C                  THE LENGTH OF THE DIAM ARRAY IS ALWAYS ONE GREATER
C                  THAN THE NUMBER OF SIZE CLASSES(MICROMETERS)
C     DMEAN        MEDIAN DIAMETER (MICROMETERS) OF LOGNORMAL PARTICLE
C                  SIZE DISTRIBUTION
C     DNS          FALLOUT PARTICLE DENSITY (GM/CM**3)
C     EXPO         EXPONENT OF THE FREQUENCY FUNCTION FOR THE POWER
C                  LAW PARTICLE SIZE FREQUENCY DISTRIBUTION
C     FMASS(I)     ARRAY OF FRACTION OF TOTAL PARTICULATE MASS IN I-TH
C                  PARTICLE SIZE CLASS. MAXIMUM LENGTH OF ARRAY = 200
C     HEIGHT       HEIGHT OF BURST (METERS) ABOVE GROUND ZERO
C     IDISTR       CONTROL INTEGER FOR PARTICLE SIZE DISTRIBUTION
C                  1 - LOGNORMAL DISTRIBUTION
C                  2 - POWER LAW DISTRIBUTION
C                  3 - TABULAR DISTRIBUTION READ IN ON CARDS (ARRAY WHY)
C     IS           CONTROL INTEGER SPECIFIES WHETHER LOGNORMAL
C                  DISTRIBUTION IS SPECIFIED BY THE USER OR BY THE
```

```
C                         PROGRAM                                                LINK1 59
C                         0 - PROGRAM SPECIFIED LOG-NORMAL DISTRIBUTION          LINK1 60
C                         1 - USER SPECIFIED LOG-NORMAL DISTRIBUTION             LINK1 61
C     ISIN                SYSTEM INPUT TAPE                                      LINK1 62
C     ISOUT               SYSTEM OUTPUT TAPE                                     LINK1 63
C     NDSTR               NUMBER OF PARTICLE SIZE CLASSES. MAXIMUM = 200         LINK1 64
C     NHODO               NUMBER OF ENTRIES IN WIND HODOGRAPH TABLE              LINK1 65
C     PS(I)               ARRAY(200), PARTICLE SIZE CLASS MIDPOINT DIAMETER      LINK1 66
C                         (METERS)                                               LINK1 67
C     SD                  STANDARD DEVIATION OF LOGNORMAL PARTICLE SIZE          LINK1 68
C                         DISTRIBUTION(DIMENSIONLESS)                            LINK1 69
C     SSAM                MASS OF CONDENSED PHASE MATERIAL AT SPECIFICATION      LINK1 70
C                         TIME                                                   LINK1 71
C     TME                 TIME OF INITIAL CONDITIONS SPECIFICATION               LINK1 72
C     TMP1                AVERAGE TEMPERATURE OF GAS IN CLOUD                    LINK1 73
C     TMP2                AVERAGE TEMPERATURE OF CONDENSED PHASE MATERIAL IN     LINK1 74
C                         CLOUD                                                  LINK1 75
C     T2M                 TEMPORARY STORAGE                                      LINK1 76
C     USOIL               SOIL CLASS INDICATOR                                   LINK1 77
C                         1.0 FOR SILICEOUS                                      LINK1 78
C                         2.0 FOR CALCAREOUS                                     LINK1 79
C     VPR                 MASS OF VAPOR IN CLOUD AT SPECIFICATION TIME           LINK1 80
C     VX(I)               X-COMPONENT OF WIND VELOCITY AT WIND HODOGRAPH         LINK1 81
C                         STRATUM I (METERS/SEC.)                                LINK1 82
C     VY(I)               Y-COMPONENT OF WIND VELOCITY AT WIND HODOGRAPH         LINK1 83
C                         STRATUM I (METERS/SEC.)                                LINK1 84
C     W                   WEAPON YIELD (KT)                                      LINK1 85
C     ZSCL                SCALED HEIGHT OF BURST                                 LINK1 86
C     ZV(I)               ALTITUDE OF THE WIND COMPONENTS VX(I) AND VY(I)        LINK1 87
C                         (METERS  RELATIVE TO MEAN SEA LEVEL)                   LINK1 88
C                                                                                LINK1 89
C ******************************************************************************LINK1 90
      COMMON /SET1/                                                              LINK1 91
     1CAY        ,DETID(12) ,DIAM(201) ,DMEAN      ,DNS        ,EXPO       ,LINK1 92
     2FMASS(200),IDISTR     ,IEXEC     ,IRISE      ,ISIN       ,ISOUT      ,LINK1 93
     3NDSTR      ,PS(200)   ,SD        ,SSAM       ,TME        ,TMP1       ,LINK1 94
     4TMP2       ,T2M       ,USOIL     ,VPR        ,W          ,HEIGHT     ,LINK1 95
     5ZSCL       ,NHODO     ,ZV(200)   ,VX(200)    ,VY(200)                 LINK1 96
C                                                                                LINK1 97
C                                                                                LINK1 98
C ******************************************************************************LINK1 99
C                                                                                LINK1100
 1    FORMAT(12A6)                                                               LINK1101
 2    FORMAT(/3X,60HTHE SPECIFIED STANDARD DEVIATION IS NEGATIVE HENCE ILINK1102
     1NCORRECT///)                                                               LINK1103
 3    FORMAT(7F10.3)                                                             LINK1104
 4    FORMAT(///25X28H****  INPUT PARAMETERS  ****/20X,5HYIELD,40X,E12.5LINK1105
     1,2X,2HKT/20X,24HHEIGHT OR DEPTH OF BURST,21X,E12.5,2X,6HMETERS/20XLINK1106
     2,13HSOIL CATEGORY)                                                         LINK1107
 5    FORMAT(1H+,65X,9HSILICEOUS)                                                LINK1108
 6    FORMAT(1H+,65X,10HCALCAREOUS)                                              LINK1109
 7    FORMAT(//20X,              36HPARTICLE SIZE FREQUENCY DISTRIBUTION/LINK1110
     125X32HA LOG-NORMAL DISTRIBUTION WITH -/30X,15HMEDIAN DIAMETER,20 ,LINK1111
     2E12.5,2X,11HMICROMETERS/30X,28HGEOMETRIC STANDARD DEVIATION, 7X,  LINK1112
     3E12.5/25X, 34HTHIS DISTRIBUTION WAS SPECIFIED BY)                          LINK1113
 8    FORMAT(1H+,65X,11HTHE PROGRAM)                                             LINK1114
 9    FORMAT(1H+,65X,8HTHE USER)                                                 LINK1115
 10   FORMAT(I5)                                                                 LINK1116
```

```
11    FORMAT(/3X,58HTHE SCALED DEPTH OF BURST IS BEYOND THE SCOPE OF THELINK1117
     1 MODEL///)                                                           LINK1118
12    FORMAT(/3X,111HTHE SCALED HEIGHT OF BURST IS SUCH THAT THERE IS NOLINK1119
     1 SOIL MASS ENTRAINED IN THE CLOUD AND HENCE NO LOCAL FALLOUT///)    LINK1120
13    FORMAT(///25X37H****  INITIAL CLOUD PROPERTIES AT H +E12.5,14H SECLINK1121
     1ONDS  ****//20X,23HAVERAGE GAS TEMPERATURE38X,E12.5,2X,14HDEGREES LINK1122
     2KELVIN//20X,56HAVERAGE TEMPERATURE OF CONDENSED PHASE MATERIAL IN LINK1123
     3CLOUD,5X,E12.5,2X,14HDEGREES KELVIN//20X,31HMASS OF VAPORIZED SOILLINK1124
     4 IN CLOUD,30X,E12.5,2X,9HKILOGRAMS//20X41HMASS OF CONDENSED PHASE LINK1125
     5MATERIAL IN CLOUD,20X,E12.5,2X,9HKILOGRAMS)                         LINK1126
15    FORMAT(1X,14HLEAVING LINK 1)                                         LINK1127
 16 FORMAT(≠1≠///51X,≠* * * * * * * * * * *≠//12X,≠T H E   D E P A R T MLINK1128
     1 E N T   O F   D E F E N S E   F A L L O U T   P R E D I C T I O NLINK1129
     2   S Y S T E M≠//51X,≠* * * * * * * * * * *≠////43X,≠INITIAL CONDITILINK1130
     3ONS (FIREBALL) MODULE≠,///55X,≠PREPARED BY≠/46X,≠MT. AUBURN RESEARLINK1131
     4CH ASSOCIATES≠/55X,≠NEWTON, MASS.≠////25X,≠*****  INITIAL CONDITIONLINK1132
     5S IDENTIFICATION  ****≠/25X,12A6)                                   LINK1133
17    FORMAT(/3X,60HTHE SPECIFIED MEAN PARTICLE SIZE IS NEGATIVE HENCE ILINK1134
     1NCORRECT///)                                                         LINK1135
 18 FORMAT(//20X,36HPARTICLE SIZE FREQUENCY DISTRIBUTION/                 LINK1136
     125X41HA TABULATED EMPIRICAL DISTRIBUTION WITH -/30XI2,2X,21HPARTICLINK1137
     2LE SIZE CLASSES)                                                     LINK1138
192   FORMAT(//51X,19H* * * * * * * * * *//)                               LINK1139
 193 FORMAT(≠1≠,9X,≠PARTICLE SIZE, LOWER SIZE INTERVAL BOUNDARY,MASS FRLINK1140
     1EQUENCY, AND UPPER SIZE INTERVAL BOUNDARY≠/10X,≠FOR USE IN PSXP,TRLINK1141
     2ANSPORT, AND ACTIVITY CALCULATIONS (DIAMETERS IN METERS)≠//20X,≠DILINK1142
     3AMETER≠,4X,≠LOWER BOUNDARY≠,3X,≠MASS FRACTION≠,3X,≠UPPER BOUNDARY≠LINK1143
     4//)                                                                  LINK1144
 197 FORMAT(//20X,≠PARTICLE SIZE FREQUENCY DISTRIBUTION≠/25X,≠POWER LAWLINK1145
     1 DISTRIBUTION WITH - ≠/30X,I3,1X,≠PARTICLE SIZE CLASSES≠/30X,≠THE LINK1146
     2SPECIFIED PARAMETERS ARE≠/30X,≠CAY =≠,2X,E12.5/30X,≠EXPO =≠,2X,E12LINK1147
     3.5)                                                                  LINK1148
 194 FORMAT(12X,I3,4(3X,E12.5))                                            LINK1149
195   FORMAT(2E12.5)                                                       LINK1150
 198 FORMAT(/3X,58HTHE PARTICLE SIZE DISTRIBUTION TABLE IS IMPROPERLY OLINK1151
     2RDERED///)                                                           LINK1152
C                                                                          LINK1153
C *************************************************************************LINK1154
C *************************************************************************LINK1155
C                                                                          LINK1156
C     READ INITIAL CONDITIONS RUN IDENTIFIER                               LINK1157
      READ (ISIN,1)(DETID(J),J=1,12)                                       LINK1158
C                                                                          LINK1159
C     WRITE OVERALL TITLE                                                  LINK1160
      WRITE (ISOUT,16)(DETID(J),J=1,12)                                    LINK1161
 20   READ(ISIN,10)IDISTR                                                  LINK1162
      READ(ISIN,10)NDSTR                                                   LINK1163
      IF(NDSTR)401,401,402                                                 LINK1164
 401 NDSTR=100                                                             LINK1165
 402 GO TO (210,220,211),IDISTR                                            LINK1166
 210 READ(ISIN,3)W,HEIGHT,USOIL,DMEAN,SD,DNS                               LINK1167
C     WAS A PRESHOT PARTICLE LOG-NORMAL DISTRIBUTION SPECIFIED BY          LINK1168
C     THE USER   YES TO 22                                                 LINK1169
      IF(DMEAN)21,21,22                                                    LINK1170
 21   IS=0                                                                 LINK1171
      GO TO 23                                                             LINK1172
 22   IS=1                                                                 LINK1173
      GO TO 23                                                             LINK1174
```

```
  220 READ(ISIN,3)W,HEIGHT,USOIL,EXPO,CAY,DNS
      GO TO 23
  211 READ(ISIN,3)W,HEIGHT,USOIL,DNS
      READ(ISIN,195)(DIAM(I),FMASS(I),I=1,NDSTR)
      LD=NDSTR+1
      READ(ISIN,195)DIAM(LD)
C
C     CHECK ORDERING OF THE HISTOGRAM TABLE
      DO 215 I=2,LD
      IF(DIAM(I) .LT. DIAM(I-1)) GO TO 215
      WRITE( ISOUT,198)
      GO TO 190
  215 CONTINUE
C
C 23  CONVERT HOB - DOB FROM METERS TO FEET
   23 HEIGHT=HEIGHT/0.3048
C     ZSCL IS THE SCALED HOB  -  DOB
   60 ZSCL=HEIGHT/((W)**(1.0/3.4))
C
C     TEST THE DATA TO SEE IF THE MODEL IS APPROPRIATE
      IF(HEIGHT)66,66,63
   63 IF(ZSCL-180.0)70,70,150
   66 IF(ZSCL+20.0)143,70,70
 70   CALL TIME
      CALL TEMP
      CALL MASS
      CALL VAPOR
      GO TO (90,95,95),IDISTR
C
C     TEST FOR ACCEPTABLE SPECIFICATIONS OF PRE-SHOT PARTICLE SIZE
C     FREQUENCY DISTRIBUTION.
 90   IF(SD)91,92,92
 91   WRITE (ISOUT,2)
      GO TO 190
   92 IF(DMEAN)94,95,95
 94   WRITE (ISOUT,17)
      GO TO 190
C
 95   CALL DSTBN
C
C     CONVERT HOB - DOB BACK TO METERS FROM FEET
      HEIGHT=HEIGHT*0.3048
C
C     CONVERT VPR AND SSAM FROM GRAMS TO KILOGRAMS
      VPR=VPR/1000.0
C     DURING COMPUTATION SSAM CONTAINS THE VALUE OF THE TOTAL MASS OF
C     GAS AND CONDENSED PHASE MATERIAL IN THE CLOUD.
      SSAM=SSAM/1000.0-VPR
C
C     WRITE INITIAL CONDITIONS RESULTS
      WRITE(ISOUT,4)W,HEIGHT
      IF(USOIL-1.0)301,301,302
 301  WRITE (ISOUT,5)
      GO TO 305
 302  WRITE (ISOUT,6)
  305 GO TO (309,310,311),IDISTR
  309 WRITE(ISOUT,7)DMEAN,SD
      IF (IS)102,103,102
```

```
 103  WRITE (ISOUT,8)
      GO TO 315
 102  WRITE (ISOUT,9)
      GO TO 315
  311 WRITE(ISOUT,198)NDSTR
C
C     PRINT FINAL PARTICLE SIZE CLASS
C
  315 WRITE(ISOUT,193)
      DO 602 J=1,NDSTR
      JO=J+1
      DM1=DIAM(JO)*1.0E-6
      DM2=DIAM(J)*1.0E-6
  602 WRITE(ISOUT,194)J,PS(J),DM1,FMASS(J),DM2
      GO TO 106
  310 WRITE(ISOUT,197)NDSTR,CAY,EXPO
      GO TO 315
 106  WRITE(ISOUT,13)TME,TMP1,TMP2,VPR,SSAM
 118  WRITE(ISOUT,192)
  200 CALL SHWIND
      WRITE(ISOUT,15)
      RETURN
 143  WRITE (ISOUT,11)
      GO TO 190
 150  WRITE (ISOUT,12)
 190  CALL EXIT
      END
```

```
      SUBROUTINE DSTBN
      COMMON /SET1/
     1CAY        ,DETID(12)  ,DIAM(201)  ,DMEAN      ,DNS        ,EXPO       ,
     2FMASS(200) ,IDISTR     ,IEXEC      ,IRISE      ,ISIN       ,ISOUT      ,
     3NDSTR      ,PS(200)    ,SD         ,SSAM       ,TME        ,TMP1       ,
     4TMP2       ,T2M        ,USOIL      ,VPR        ,W          ,HEIGHT     ,
     5ZSCL       ,NHODO      ,ZV(200)    ,VX(200)    ,VY(200)
C
C     LOGNORMAL DISTRIBUTION TO 100
C     POWER FUNCTION DISTRIBUTION TO 200
C     TABULAR DISTRIBUTION TO 300
C
C     EQUATION 26.2.23 OF NBS-AMS 55 HANDBOOK IS USED TO COMPUTE THE
C     PROBABILITY FUNCTION ARGUMENT FROM THE RATIONAL POLYNOMIAL
C     APPROXIMATION TO THE NORMAL PROBABILITY FUNCTION.
      TA(X)=SQRT(ALOG(1.0/X**2))
      APX(X)=TA(X)-(2.515517+0.802853*TA(X)+0.010328*TA(X)**2)/
     1(1.0+1.432788*TA(X)+0.189269*TA(X)**2+0.001308*TA(X)**3)
      LD=NDSTR+1
      GO TO (100,200,300),IDISTR
  100 IF(DMEAN)111,111,112
  111 DMEAN=0.407
      SD=4.0
  112 IF(NDSTR-1)101,101,102
  101 PS(1)=DMEAN*1.0E-6
      C5=SD**5
      DIAM(1)=DMEAN*C5
      DIAM(2)=DMEAN/C5
      FMASS(1)=1.0
      GO TO 400
  102 BARMU=ALOG(DMEAN)
      SIGMA=ALOG(SD)
      BARMU=BARMU+3.*SIGMA**2
      FRAC=1.0/FLOAT(NDSTR)
      DO 103 ND=1,NDSTR
  103 FMASS(ND)=FRAC
      NH=NDSTR/2
      DO 104 I=1,NH
      PRB=FLOAT(I)*FRAC
      DIAM(I+1)=BARMU+APX(PRB)*SIGMA
      J=NDSTR-I+1
  104 DIAM(J)=BARMU-APX(PRB)*SIGMA
C
C     FOR THE 2 EXTREME INTERVALS THE AVERAGE DIAMETER IS
C     ASSUMED TO BE AT HALF A MASS FRACTION FROM ZERO AND ONE
C
      PRB=FRAC/2.0
      PS(1)=BARMU+APX(PRB)*SIGMA
      PS(NDSTR)=BARMU-APX(PRB)*SIGMA
      DIAM(1)=2.*PS(1)-DIAM(2)
      DIAM(LD)=2.*PS(NDSTR)-DIAM(NDSTR)
C
C     CALCULATE MEAN DIAMETERS FROM BOUNDARY VALUES
C
      J=NDSTR-1
      IF(J-1)107,107,105
  105 DO 106 I=2,J
  106 PS(I)=0.5*(DIAM(I)+DIAM(I+1))
```

```
  107 DO 108 I=1,NDSTR
      DIAM(I)=EXP(DIAM(I))
  108 PS(I)=EXP(PS(I))*1.0E-6
      DIAM(LD)=EXP(DIAM(LD))
      GO TO 400
  200 IF(EXPO-4.0)201,202,202
  202 WRITE(ISOUT,2001)
      CALL EXIT
 2001 FORMAT(≠1≠,1X,≠EXPONENT OF POWER LAW POWER LAW PARTICLE SIZE FREQU
     1ENCY DISTRIBUTION GT. OR EQ. 4.0≠)
  201 IF(NDSTR-1)203,204,204
  203 NDSTR=10
  204 AN=FLOAT(NDSTR)
      FRAC=1.0/AN
      DO 205 I=1,NDSTR
  205 FMASS(I)=FRAC
      POW=1.0/(4.0-EXPO)
      DMIN=(6.0*SSAM*FRAC/(POW*CAY*DNS*3.14159E6))**POW
      DO 206 IJ=1,NDSTR
      AJ=FLOAT(IJ)-1.0
  206 DIAM(IJ)=(AN-AJ)**POW*DMIN
      PS(NDSTR)=DMIN*0.5**POW
      DIAM(LD)=PS(NDSTR)**2/DIAM(NDSTR)
      ND=NDSTR-1
      DO 207 IJ=1,ND
  207 PS(IJ)=SQRT(DIAM(IJ)*DIAM(IJ+1))
      DO 208 IJ=1,LD
  208 DIAM(IJ)=1.0E+6*DIAM(IJ)
      GO TO 400
  300 DO 301 I=1,NDSTR
  301 PS(I)=0.5*(DIAM(I)+DIAM(I+1))*1.0E-6
  400 RETURN
      END
```

```
      SUBROUTINE SHWIND
C
C     READS IN SHOT TIME WIND DATA ABOVE GROUND ZERO
C
      COMMON /SET1/
     1CAY       ,DETID(12) ,DIAM(201) ,DMEAN     ,DNS       ,EXPO      ,
     2FMASS(200),IDISTR    ,IFXEC     ,IRISE     ,ISIN      ,ISOUT     ,
     3NDSTP     ,PS(200)   ,SD        ,SSAM      ,TME       ,TMP1      ,
     4TMP2      ,T24       ,USOIL     ,VPR       ,W         ,HEIGHT    ,
     5ZSCL      ,NHODO     ,ZV(200)   ,VX(200)   ,VY(200)
      READ(ISIN,1)NHODO
      IF(NHODO)100,100,200
  100 WRITE(ISOUT,5)
      GO TO 300
  200 READ(ISIN,2)(ZV(J),VX(J),VY(J),J=1,NHODO)
      WRITE(ISOUT,3)NHODO
      WRITE(ISOUT,4)(ZV(J),VX(J),VY(J),J=1,NHODO)
  300 RETURN
C
    1 FORMAT(I5)
    2 FORMAT(F12.3,     2F12.3)
    3 FORMAT(≠1≠,9X,≠WIND HODOGRAPH AT GROUND ZERO≠,10X,≠NHODO = ≠,I5//1
     11X,≠VECTOR ALTITUDE, ZV(J)≠,16X,≠VX(J)≠,24X,≠VY(J)≠)
    4 FORMAT(3(16X,E13.6))
    5 FORMAT(≠1≠,9X,≠SHOT-TIME WINDS HAVE NOT BEEN SPECIFIED≠)
      END
```

REFERENCES

1. H. G. Norment, W. Y. G. Ing, and J. Zuckerman, "Department of Defense Land Fallout Prediction System. Volume II. Initial Conditions," DASA-1800-II, Technical Operations Research, Inc., Report TO-B 66-44 (30 September 1966). AD 803-144L.

2. H. G. Norment and S. Woolf, "Department of Defense Land Fallout Prediction System. Volume III. Cloud Rise (Revised)," DASA-1800-III (Revised), Arcon Corp. Report R70-1W (1 September 1970).

3. J. Aitchison and J. A. C. Brown, <u>The Lognormal Distribution</u> (Cambridge University Press, Cambridge, England, 1966).

4. M. Abramowitz and I. A. Stegun, <u>Handbook of Mathematical Functions with Formulas, Graphs, and Mathematical Tables</u>, National Bureau Standards Applied Mathematics Series 55 (U.S.G.P.O., June 1964).

5. E. C. Freiling, "A Comparison of the Fallout Mass-Size Distributions Calculated by Lognormal and Power-Law Models," USNRDL-TR-1105 (14 November 1966). AD 646-019.